# The Birthday Problem

Problem: In a group of $n$ randomly chosen people, what is the probability that at least 2 of them share the same birthday?

The problem can be tackled first by theory. If $A$ is the event that at least two of the people share the same birthday, then $A'$ is the event that no two people share the same birthday. It is easier to calculate Pr($A'$).

There are $365^n$ ways of selecting n random birthdays with repetitions possible. The number of ways of selecting n unique birthdays is $365 \times 364 \times 363 \times \cdots \times (365 - n + 1)$ because there are 365 choices for the first birthday, but only 364 choices for the second birthday, 363 for the third, and so on.
Therefore $\Pr(A') = \dfrac{365 \times 364 \times 363 \times \cdots \times (365 - n + 1)}{365^n}$. So the probability we are looking for is $\Pr(A) = 1 - \dfrac{365 \times 364 \times 363 \times \cdots \times (365 - n + 1)}{365^n}$.

In[ ]:=
```
bdaysolve[n_] := N[1 - (Product[365 - i, {i, 0, n - 1}]) / (365^n)]
```

In[ ]:=
```
bdaysolve[23]
```

Out[ ]=
```
0.507297
```

In[ ]:=
```
bdaysim[n_] := Length[DeleteDuplicates[RandomInteger[{1, 365}, n]]] ≠ n
```

In[ ]:=
```
trials = 1 000 000;
```

In[ ]:=
```
bdaytry[n_] := N[Count[Table[bdaysim[n], trials], True] / trials]
```

In[ ]:=
```
bdaytry[23]
```

Out[ ]=
```
0.50762
```

In[ ]:=
```
ListPlot[Table[bdaysolve[i], {i, 1, 365}]]
```

Out[ ]=



In[ ]:=

# Lagrange’s Four Square Theorem

It can be proved that every positive integer can be expressed as the sum of four or fewer square numbers. For example, $77 = 8^2 + 3^2 + 2^2$.

The proof: it is enough to prove that every prime number is the sum of four or fewer squares due to the following identity: $\left(a^2+b^2+c^2+d^2\right)\left(A^2+B^2+C^2+D^2\right) = (a\,A + b\,B + c\,C + d\,D)^2 +$

$$(a\,B - b\,A - c\,D + d\,C)^2 + (a\,c + b\,D - c\,A - d\,B)^2 + (a\,D - b\,C + c\,B - d\,A)^2$$

The proof uses a technique called *descent*. The idea is to show that a multiple of a prime number can be expressed as the sum of four squares, and from that multiple deduce a smaller multiple that can also be expressed as the sum of four squares. Eventually we will reach the first multiple of the prime.

In[ ]:=
```
reduce[a_, p_] := Module[{m}, m = Mod[a, p]; If[m > p / 2, Return[m - p], Return[m]]]
```

In[ ]:=
```
fibIdentFour[a1_, a2_] := {a * A + b * B + c * C + d * D,
    a * B - b * A - c * D + d * C, a * C + b * D - c * A - d * B, a * D - b * C + c * B - d * A} /.
  {a → a1[[1]], b → a1[[2]], c → a1[[3]], d → a1[[4]], A → a2[[1]], B → a2[[2]], C → a2[[3]], D → a2[[4]]}
```

In[ ]:=
```
pFourSquares[p_] :=
 Module[{start, i, m, a, b, c, d, A, B, C, D, r, xs, ys, target, P}, start = None;
  P = (p - 1) / 2;
  xs = Table[Mod[i^2, p], {i, 0, P}];
  ys = Table[Mod[-1 - i^2, p], {i, 0, P}];
  target = Intersection[xs, ys][[1]];
  start = {Position[xs, target][[1]][[1]] - 1, Position[ys, target][[1]][[1]] - 1, 1, 0};
  m = (start[[1]]^2 + start[[2]]^2 + 1) / p;
  While[m > 1, a = start[[1]];
   b = start[[2]];
   c = start[[3]];
   d = start[[4]];
   A = reduce[a, m];
   B = reduce[b, m];
   C = reduce[c, m];
   D = reduce[d, m];
   r = (A^2 + B^2 + C^2 + D^2) / m;
   start = fibIdentFour[{a, b, c, d}, {A, B, C, D}] / m;
   m = r];
  Return[start]]
```

In[ ]:=
```
pPowerFourSquares[pe_] := Module[{p, e, i, res, psquare}, p = pe[[1]];
  e = pe[[2]];
  If[p == 2, psquare = {1, 1, 0, 0}, psquare = pFourSquares[p]];
  If[psquare == None, Print["Shouldn't happen"];];
  res = psquare;
  For[i = 2, i ≤ e, i++, res = fibIdentFour[res, psquare]];
  Return[res]]
```

In[ ]:=
```
fourSquares[n_] := Module[{factors, res, i, tmp}, factors = FactorInteger[n];
  res = pPowerFourSquares[factors[[1]]];
  For[i = 2, i ≤ Length[factors], i++, tmp = pPowerFourSquares[factors[[i]]];
   If[tmp == Null, Return[Null]];
   res = fibIdentFour[res, tmp]];
  Return[Abs[res]]]
```

In[ ]:=
```
fourSquares[896]
```

Out[ ]=

{8, 24, 0, 16}

In[ ]:=
```
fourSquares[3 843 249]
```

Out[ ]=

{1162, 1131, 1100, 62}

In[ ]:=
```
Total[%^2]
```

Out[ ]=

3 843 249